

MATH 2270-2
MAPLE PROJECT 2b - Inner product spaces
 October 22, 2001

We will explore two interesting inner product spaces. Refer to section 5.5 of the text.

A good dot product and interval for orthogonal polynomials:

The first inner product and interval we mention are used in numerical analysis algorithms. We'll call the inner product "dot1". It is the inner product of example 7, page 231.

```
[ > restart:with(plots):
  > dot1:=(f,g)->int(f(t)*g(t),t=-1..1);
      #our first inner product, for the
      #interval -1<t<1
      dot1 := (f, g) → ∫-11 f(t)g(t) dt
  > dot1(t->t,t->1);
      #the dot product of f(t)=t with g(t)=1
      #you should get zero. Meaning what?
```

Meaning you just showed that $f(t)=t$ and $g(t)=1$ are orthogonal for the dot1 inner product! Does this surprise you? In fact, once the inner product is defined, you can define magnitude, distance, angle between vectors, and you can even do projection problems. This is discussed in detail in section 5.5.

```
[ > mag1:=f->sqrt(dot1(f,f)):
      # the magnitude of a vector

  > dist1:=(f,g)->mag1(f-g):
      # the ``distance`` between two functions
  > cosangle1:=(f,g)->(dot1(f,g)/(mag1(f)*mag1(g))):
      #computes the "cos of the angle" between functions
  > fangle1:=(f,g)->evalf(arccos(cosangle1(f,g))):
      #computes angle between functions
```

For example we may use Gram-Schmidt to find an orthonormal basis for the polynomial subspace $P_2 = \text{span}\{1, t, t^2\}$:

```
[ > P0:=t->1;
      P1:=t->t;
      P2:=t->t^2;
      #our usual "natural" basis
      P0 := 1
      P1 := t → t
      P2 := t → t2
```

```

> W0:=P0/mag1(P0):
  #first orthonormal vector
Z1:=P1-dot1(P1,W0)*W0:
  #P1 was orthogonal to P0, so don't really
  #need the usual projection formula, but
  #here it is anyway.
W1:=Z1/mag1(Z1):
  #second orthonormal vector
Z2:=P2-dot1(P2,W0)*W0-dot1(P2,W1)*W1:
W2:=Z2/mag1(Z2):
  #third orthonormal vector
> W0(t);W1(t);W2(t);
  #the orthonormal polynomials we get

```

$$\frac{1}{2}\sqrt{2}$$

$$\frac{1}{2}t\sqrt{6}$$

$$\frac{3}{4}\left(t^2-\frac{1}{3}\right)\sqrt{10}$$

Projection:

Now that we have an orthonormal basis for P2 we can do projection problems. We will try to find the closest degree 2 polynomial to $f(t)=\exp(t)$, using our inner product dot1 to measure distance:

```

[ > proj2:=f->evalf(dot1(f,W0)*W0+dot1(f,W1)*W1+dot1(f,W2)*W2);
  #this is the usual projection formula, but we use evalf to
  #get decimals rather than messy algebraic numbers.
[ > f:=t->exp(t);
[ > g:=t->proj2(f)(t);

```

This polynomial g is "closer" to $\exp(t)$ than the usual Taylor polynomial $p(t)=1+t+t^2/2$, when we use the distance which we get from dot1. We can compare distances:

```

[ > p:=t->1+t+t^2/2;
[ > evalf(dist1(f,g));
  evalf(dist1(f,p));
[


```

So, at least for our distance, the function g does about three times as well as the Taylor polynomial. You can also see this geometrically, by plotting the three graphs on the interval from -1 to 1:

```

[ > with(plots):
[ > actual:=plot(f(t),t=-1..1,color=red):
  best:=plot(g(t),t=-1..1,color=black):
  tayl:=plot(p(t),t=-1..1,color=blue):
  display({actual,best,tayl});

```



The orthogonal polynomials which you have been constructing are called the Legendre polynomials. Maple knows about them. To see the first few you can load the ‘orthopoly’ package. By the way, if you define different weighted inner products you get different (famous to experts) orthogonal polynomial families, you can read about some of them on the help windows, starting at ‘orthopoly’. Orthogonal polynomials are used in approximation problems, as you might expect.

```
[ > with(orthopoly):
[ > P(0,x);P(1,x);P(2,x);P(3,x);P(4,x);P(5,x);
[     #these should look familiar, after what you just did!
[     #they haven't been normalized, though.
[ >
```

Fourier Series:

Here’s an interval and dot product which makes the usual trig functions into an orthonormal family! See page 233 of the text!

```
[ > dot2 := (f,g) -> 1/Pi * int(f(t)*g(t), t=-Pi..Pi);
```

$$dot2 := (f, g) \rightarrow \frac{\int_{-\pi}^{\pi} f(t)g(t) dt}{\pi}$$

[For example:

```
[ > f:=t->sin(5*t);
[     g:=t->cos(3*t);
[     dot2(f,f);
[     dot2(g,g);
[     dot2(f,g);
```

We add the constant function to our collection, but it will need to be normalized:

```
[ > dot2(1,1);
  dot2(f,1);
  dot2(g,1);
  #the constant function has norm squared
  #equal to 2, however. But it is orthogonal
  #to all cos(kt),sin(kt), k a natural number
```

Thus, for any n , the family $\{1/\sqrt{2}, \cos(t), \cos(2t), \dots, \cos(nt), \sin(t), \sin(2t), \dots, \sin(nt)\}$ is an orthonormal basis of a $2n+1$ dimensional subspace of functions. So it is easy to project onto this subspace using our usual projection formulas. It is an amazing fact that if $f(t)$ is any piecewise continuous function on the interval $-\pi \leq t \leq \pi$, then as n approaches infinity the distances between these projections and f converges to zero. Here's an example, see page 236:

```
[ > f:=t->t:
  #the function we shall decompose into trigonometric pieces
[ > a0:=int(f(t),t=-Pi..Pi);
  #could you have predicted the answer?
  a0:=0
[ > #now get projection coefficients
  #AKA Fourier coefficients
  for i from 1 to 10 do
  g1:=t->cos(i*t):
  g2:=t->sin(i*t):
  a[i]:=dot2(g1,f):
  b[i]:=dot2(g2,f):
  od:
[ >
[ > fsum:=t-> a0/(2*Pi)+ sum(a[j]*cos(j*t),j=1..10) +
  sum(b[j]*sin(j*t),j=1..10);
[ > fsum(t);
  #why are there no cosine terms?
[ > plot1:=plot(fsum(t),t=-Pi..Pi,color=black):
  plot2:=plot(t,t=-Pi..Pi,color=red):
  display({plot1,plot2});
```

There are problems for you to do, on the template 2270proj2b.mws, which you can download from our Maple page.